

Массивы.

ОПЕРАЦИИ С МАССИВАМИ

- Часто при работе с таблицами возникает необходимость применить одну и ту же операцию к целому диапазону ячеек или произвести расчеты по формулам, зависящим от большого массива данных.
- Под массивом в **MS Excel** понимается прямоугольный диапазон формул или значений, которые программа обрабатывает как единую группу.

1. Умножение элементов массива на число

- В качестве примера использования формулы массива приведем расчет цен группы товаров с учетом **НДС** (налог на добавленную стоимость).
- Пусть в диапазоне **B2:B4** даны цены группы товаров без учета НДС. Необходимо найти цену каждого товара с учетом НДС, который будем полагать равным **25%**.

	А	В	С
	Товар	Цена без НДС	Цена с учетом НДС
1			
2	Телевизор	400	
3	Компьютер	920	
4	Холодильник	1000	

- Таким образом, необходимо умножить массив элементов **B2:B4** на **125%**. Результат надо разместить в ячейках диапазона **C2:C4**. Для этого:
- следует выделить диапазон **C2:C4**, ввести формулу
=B2:B4*125%
- завершить ввод формулы не нажатием клавиши **<Enter>**, а нажатием комбинации клавиш **<Ctrl>+<Shift>+<Enter>**.
- Таким образом, вы сообщите **MS Excel**, что необходимо выполнить операцию над массивом, т. е. создать формулу массива.
- В ответ **MS Excel** автоматически возьмет формулу в фигурные скобки:

{=B2:B4*125%}

Умножение элементов массива на число

C2		fx {=B2:B4*125%}	
	A	B	C
	Товар	Цена без НДС	Цена с учетом НДС
1			
2	Телевизор	400	500
3	Компьютер	920	1150
4	Холодильник	1000	1250

2. Поэлементное сложение, вычитание, умножение и деление двух массивов

- Рассмотрим операцию поэлементного сложения двух массивов.
- Пусть, например, слагаемыми будут массивы, содержащиеся в диапазонах **A2:B3** и **D2:E3**.

	A	B	C	D	E
1	Массив A			Массив B	
2	4	1		6	7
3	6	10		1	5

	A	B	C	D	E
1	Массив А			Массив В	
2	4	1		6	7
3	6	10		1	5

- Необходимо выделить на рабочем листе диапазон, например, **G2:H3**, в который будет помещен результат поэлементного сложения двух массивов. От данного диапазона требуется, чтобы он имел **тот же размер, что и массивы-слагаемые**.
- Далее следует ввести формулу

=A2:B3+D2:E3

- Завершить ввод следует нажатием комбинации клавиш

<Ctrl>+<Shift>+<Enter>.

- **MS Excel** возьмет формулу в строке формул в фигурные скобки и произведет требуемые вычисления:

{=A2:B3+D2:E3}

G2		fx {=A2:B3+D2:E3}						
	A	B	C	D	E	F	G	H
1	Массив A			Массив B			Массив C	
2	4	1		6	7		10	8
3	6	10		1	5		7	15

- Аналогично можно вычислить поэлементно **разность, произведение и деление массивов.**
- Для избежания ошибок в формулу вводите ссылки на диапазоны ячеек не с клавиатуры, а путем выбора их на рабочем листе мышью. Тогда ссылка на диапазон ячеек в формулу будет вводиться автоматически.

3. Вычисление сложных выражений

- Приведем более сложный пример использования формул массива, а именно: попытаемся найти значение следующего выражения:

$$S = \frac{2 \sum_{i=1}^n X_i + \left(\sum_{i=1}^m \sum_{j=1}^m B_{ij} C_{ij} \right)^2}{1 + \sum_{i=1}^n X_i^2}$$

- где X – вектор из n компонентов, B и C – матрицы размера $m \times m$, причем, $n = 3$, $m = 2$

$$X = \begin{pmatrix} 2 \\ 1 \\ 2 \end{pmatrix}, B = \begin{pmatrix} 2 & 1 \\ 2 & 5 \end{pmatrix}, C = \begin{pmatrix} 1 & 1 \\ 4 & 5 \end{pmatrix}$$

	A	B	C	D	E
1	Массив X	Массив В		Массив С	
2	2	2	1	1	1
3	1	2	5	4	5
4	2				

Для вычисления значения **S** необходимо:

- ввести в диапазон **A2:A4** компоненты вектора **X**;
- ввести в диапазон **B2:C3** компоненты матрицы **B**;
- ввести в диапазон **D2:E3** компоненты матрицы **C**.

- ввести в ячейку **B6** следующую формулу:

$\{=(2*\text{СУММ}(A2:A4)+\text{СУММ}(B2:C3*D2:E3)^2)/(1+\text{СУММ}(A2:A4^2))\}$

	A	B	C	D	E	F	G	H
1	Массив X	Массив A		Массив B				
2	2	2	1	1	1			
3	1	2	5	4	5			
4	2							
5								
6	S=	130,6						

4. Функции для работы с матрицами

- В **MS Excel** имеются функции для работы с матрицами, перечисленные в таблице:

Функция	Описание
МОБР (массив)	Возвращает обратную матрицу
МОПРЕД (массив)	Возвращает определитель матрицы
МУМНОЖ (массив ; массив2)	Возвращает матричное произведение двух матриц
ТРАНСП (массив)	Возвращает транспонированную матрицу

Пример 1. Решение системы линейных уравнений

Решение линейной системы уравнений
 $AX = B$, где

- A – матрица коэффициентов,
- B – столбец (вектор) свободных членов,
- X – столбец (вектор) неизвестных,

имеет вид $X = A^{-1}B$, где A^{-1} – обратная матрица к A .

- Пусть: $A = \begin{pmatrix} 8 & 3 \\ 2 & 7 \end{pmatrix}, B = \begin{pmatrix} 4 \\ 2 \end{pmatrix}$
- Введем исходные данные задачи на рабочий лист **EXCEL**:

	А	В	С	Д	Е
1	Матрица коэффициентов			Свободные члены	
2	8	3		4	
3	2	7		2	

	F2	fx {=МУМНОЖ(МОБР(A2:B3);D2:D3)}				
	A	B	C	D	E	F
1	Матрица коэффициентов			Свободные члены		Решение
2	8	3		4		0,44
3	2	7		2		0,16

- Выделим тот диапазон, в который будет введено решение. Например, **F2:F3**.
- Введем в него формулу
=МУМНОЖ(МОБР(A2:B3);D2:D3)
- Завершим ввод формулы нажатием комбинации клавиш
<Ctrl>+<Shift>+<Enter>.
- **MS Excel** возьмет формулу в строке формул в фигурные скобки и произведет требуемые вычисления с элементами массива:

{=МУМНОЖ(МОБР(A2:B3);D2:D3)}

- Таким образом, решением системы уравнений является вектор

$$X = \begin{pmatrix} 0,44 \\ 0,16 \end{pmatrix}$$

F2  $\{=МУМНОЖ(МОБР(A2:B3);D2:D3)\}$						
	A	B	C	D	E	F
1	Матрица коэффициентов			Свободные члены		Решение
2	8	3		4		0,44
3	2	7		2		0,16

Пример 2. Решение системы линейных уравнений

- В качестве более сложного примера решим систему линейных уравнений

$$A^2X = B,$$

- где

$$A = \begin{pmatrix} 7 & 2 \\ 1 & 4 \end{pmatrix}, B = \begin{pmatrix} 2 \\ 1 \end{pmatrix}$$

- Решением этой системы является вектор $X = (A^2)^{-1}B$.

Для нахождения вектора **X**:

- Введем элементы матрицы **A** и вектора **B**.

	А	В	С	Д	Е
1	Матрица коэффицентов			Свободные члены	
2	7	2		2	
3	1	4		1	

- Выберем диапазон **F2:F3**, куда поместим элементы вектора решения.
- Введем в этот диапазон формулу:

	А	В	С	Д	Е
1	Матрица коэффициентов			Свободные члены	
2	7	2		2	
3	1	4		1	

=МУМНОЖ(МОБР(МУМНОЖ(А2:В3; А2:В3));D2:D3)

- Завершим ввод формулы нажатием комбинации клавиш **<Ctrl>+<Shift>+<Enter>**. **MS Excel** возьмет формулу в строке формул в фигурные скобки и произведет требуемые вычисления с элементами массива:

{=МУМНОЖ(МОБР(МУМНОЖ(А2:В3; А2:В3));D2:D3)}

- В диапазоне ячеек **F2:F3** будет найдено решение системы уравнений:

$$X = \begin{pmatrix} 0,02071 \\ 0,04289 \end{pmatrix}$$

F2  <i>f</i> {=МУМНОЖ(МОБР(МУМНОЖ(A2:B3;A2:B3));D2:D3)}						
	A	B	C	D	E	F
1	Матрица коэффициентов			Свободные члены		Решение
2	7	2		2		0,02071
3	1	4		1		0,04290

5. Нахождение значения квадратичной формы

- Рассмотрим пример вычисления квадратичной формы $Z = X^T A X$, при этом

$$A = \begin{pmatrix} 8 & 3 \\ 2 & 7 \end{pmatrix}, X = \begin{pmatrix} 4 \\ 2 \end{pmatrix}$$

Для нахождения этой квадратичной формы:

- Введем элементы матрицы **A** и вектора **X**.

	A	B	C	D
1	Матрица A			Вектор X
2	8	3		4
3	2	7		2

	А	В	С	Д
1	Матрица А			Вектор X
2	8	3		4
3	2	7		2

- Выберем ячейку **F2** для вычисления значения формы.
- Введем формулу:

**=МУМНОЖ(МУМНОЖ(ТРАНСП(D2:D3);
A2:B3);D2:D3)**

- Завершим ввод формулы нажатием комбинации клавиш **<Ctrl>+<Shift>+<Enter>**. **MS Excel** возьмет формулу в строке формул в фигурные скобки и произведет требуемые вычисления с элементами массива:

**{=МУМНОЖ(МУМНОЖ(ТРАНСП(D2:D3);
A2:B3);D2:D3)}**

- В ячейке **F2** будет получено искомое значение формы **196**.

F2		fx {=МУМНОЖ(МУМНОЖ(ТРАНСП(D2:D3);A2:B3);D2:D3)}						
	A	B	C	D	E	F	G	H
1	Матрица A			Вектор X		Квадратичная форма Z		
2	8	3		4		196		
3	2	7		2				

Примеры использования матричных операций в экономических задачах

- **Пример 1.** Данные о доходах (тыс. ден. ед.) холдинга по трем регионам трех компаний за 2008 и 2009 гг. представлены в матрицах A и B:

$$A = \begin{bmatrix} 550 & 880 & 310 \\ 2050 & 340 & 190 \\ 2500 & 220 & 700 \end{bmatrix} \text{ и } B = \begin{bmatrix} 620 & 1010 & 320 \\ 2080 & 680 & 390 \\ 2460 & 930 & 700 \end{bmatrix}$$

- Здесь элемент a_{ij} матрицы A означает доход i -й компании в j -м регионе за 2008 г.
Аналогично – для матрицы B , но за 2009 г.
- Вычислить матрицу C приростов доходов за период с 2008 по 2010 г. и проанализировать ее.
- Рассчитать матрицу $C_{\text{ср}}$, характеризующую средние размеры приростов доходов компаний холдинга за год.

Решение.

- 1) Матрица **С** приростов доходов за рассматриваемый период равна: **$C = B - A$** .

	A8	fx {=E2:G4-A2:C4}					
	A	B	C	D	E	F	G
1	Матрица A				Матрица B		
2	550	880	310		620	1010	320
3	2050	340	190		2080	680	390
4	2500	220	700		2460	930	700
5							
6							
7	Матрица C				Матрица C _{ср}		
8	70	130	10				
9	30	340	200				
10	-40	710	0				

- Элементы матрицы **С** выражают изменение доходов с 2008 по 2010 г. Так, третья компания по первому региону потерпела убытки в размере 40 тыс. ден. ед. ($c_{31} = -40$), эта же компания (третья) по третьему региону в этот же период не принесла доходов ($c_{33} = 0$).

- 2) Матрица C_{cp} , характеризующая средние размеры приростов доходов компаний холдинга за год, равна матрице C, деленной на n – количество лет в рассматриваемом периоде. В период с 2008 по 2010 г. входит 2 года (т.е. 2008 и 2009гг.), значит, $n = 2$, тогда:

$$C_{cp} = C/2.$$

E8	fx {=A8:C10/2}						
	A	B	C	D	E	F	G
1	Матрица A				Матрица B		
2	550	880	310		620	1010	320
3	2050	340	190		2080	680	390
4	2500	220	700		2460	930	700
5							
6							
7	Матрица C				Матрица C _{cp}		
8	70	130	10		35	65	5
9	30	340	200		15	170	100
10	-40	710	0		-20	355	0

Пример 2.

- В таблице приведены расценки на выполнение работ для каждого вида оборудования.
- Найдите расчетные объемы работ (количество часов использования оборудования), которые смогут окупить затраты на эксплуатацию.

Вид работ	Нормативы по видам оборудования, ден.ед.			Полные затраты на эксплуатацию, ден.ед.
	Механическое	Тепловое	Энергетическое	
Техническое обслуживание	3	1	4	85
Текущие услуги	2	2	3	82
Капитальный ремонт	10	20	15	580

Решение.

- Пусть необходимо x_1 ч работы механического оборудования, x_2 ч работы теплового оборудования и x_3 ч работы энергетического оборудования, чтобы окупить затраты на техническое обслуживание, текущие услуги и капитальный ремонт.
- Тогда из условий задачи следует система уравнений:

$$\begin{cases} 3x_1 + x_2 + 4x_3 = 85 \\ 2x_1 + 2x_2 + 3x_3 = 82 \\ 10x_1 + 20x_2 + 15x_3 = 580 \end{cases}$$

или в матричной форме **AX = B**,
где

$$A = \begin{bmatrix} 3 & 1 & 4 \\ 2 & 2 & 3 \\ 10 & 20 & 15 \end{bmatrix}, \quad B = \begin{bmatrix} 85 \\ 82 \\ 580 \end{bmatrix}, \quad X = \begin{bmatrix} x_1 \\ x_2 \\ x_3 \end{bmatrix}$$

A8		fx {=МУМНОЖ(МОБР(A2:C4);E2:E4)}								
	A	B	C	D	E	F	G	H	I	J
1	Матрица A нормативов по видам работ				Матрица полных затрат на эксплуатацию B					
2	3	1	4		85					
3	2	2	3		82					
4	10	20	15		580					
5										
6										
7	$X=A^{-1}*B$ Неизвестные объемы работ									
8	12									
9	17									
10	8									
11										

- Чтобы окупить затраты на эксплуатацию, требуется следующий объем работ:
- механическое оборудование – **12 ч** работы;
- тепловое – **17 ч**;
- энергетическое – **8 ч**.